

```

model{
  y1[1:Q, 1:Q] ~ dmulti(p1[1:Q, 1:Q], n1)
  y2[1:Q, 1:Q] ~ dmulti(p2[1:Q, 1:Q], n2)
  p1[1,1] <- pi1*eta11 + (1-pi1)*theta11
  p1[1,2] <- pi1*eta12 + (1-pi1)*theta12
  p1[2,1] <- pi1*eta21 + (1-pi1)*theta21
  p1[2,2] <- pi1*eta22 + (1-pi1)*theta22
  p2[1,1] <- pi2*eta11 + (1-pi2)*theta11
  p2[1,2] <- pi2*eta12 + (1-pi2)*theta12
  p2[2,1] <- pi2*eta21 + (1-pi2)*theta21
  p2[2,2] <- pi2*eta22 + (1-pi2)*theta22
  eta11 <- lambdaD*eta1
  eta12 <- eta1 - eta11
  eta21 <- gammaD*(1-eta1)
  eta22 <- 1 - eta11 - eta12 - eta21
  theta11 <- 1 - theta12 - theta21 - theta22
  theta12 <- gammaDc*(1-theta1)
  theta21 <- theta1 - theta22
  theta22 <- lambdaDc* theta1
  eta2 <- eta11 + eta21
  theta2 <- theta22 + theta12
  rhoD <- (eta11 - eta1*eta2) / sqrt(eta1*(1-eta1)*eta2*(1-eta2))
  rhoDc <- (theta22 - theta1*theta2) / sqrt(theta1*(1-theta1)*theta2*(1-theta2))
  pi1 ~ dbeta(5.96, 22.13) ##mode=0.19, 5th%ile=0.10, prev pop1 Mnisi
  pi2 ~ dbeta(3.44, 22.99) ##mode=0.10, 95%ile=0.25, prev pop2 Gauteng
  eta1 ~ dbeta(19.45, 3.76) ##mode=0.87, 5%ile=0.70, sens RLB
  theta1 ~ dbeta(30.92, 3.60) ##mode=0.92, 5%ile=0.80, spec RLB
  lambdaD ~ dbeta(1, 1) ## Mode=0.50, 5th%tile=0.05, sens PCR
  gammaD ~ dbeta(1, 1)
  lambdaDc ~ dbeta(1, 1) ## Mode=0.50, 5th%tile=0.05, spec PCR
  gammaDc ~ dbeta(1, 1)
}

list(n1=103, n2=154, Q=2,
y1=structure(.Data=c(
62,10,9,22),.Dim=c(2,2)),
y2=structure(.Data=c(
5,0,0,149),.Dim=c(2,2)))
list(pi1=0.19, pi2=0.10, eta1=0.87, theta1=0.92, lambdaD=0.5, lambdaDc=0.5, gammaD=0.5,
gammaDc=0.5)

```